Symbolic Summation

Victor Adamchik

Carnegie Mellon University

Wilf-Zeilberger's algorithm



The art of doing mathematics consists in finding that special case which contains all the germs of generality -

David Hilbert

Forethoughts

$F(n,\ k)$ is Gosper-summable if there is a rational function G such that

$$F(n,\ k) \ = \ G(n,\ k+1) - G(n,\ k)$$

Moreover, G is a rational multiple of F:

$$G(n,\ k) = R(n,\ k)\,F(n,\ k)$$

Indefinite summation:

$$\sum_k F(n,\ k) = \sum_k [G(n,\ k+1) - G(n,\ k)] = G(n,\ k)$$

Definite summation:

$$\sum_{k=1}^{n} F(n,\ k) = \sum_{k=1}^{n} [G(n,\ k+1) - G(n,\ k)] = G(n,\ n+1) - G(n,\ 1)$$

What do we do if the summand $F(n,\ k)$ is not Gosper-summable? Doron Zeilberger observed that Gosper's algorithm of indedfinite summation could be used in a non-obvious and nontrivial way, namely for PROVING combinatorial identities.

Definition. $F(n, k)$ has a finite support if $F(n, k) \neq 0$ only for finitely many $k \in Z$ and fixed $n \in N_0$.

In other words, all such series

$$\sum_{k=-\infty}^{\infty} F(n, k) \tag{1}$$

are actually finite sums.For example,

$$\sum_{k=-\infty}^{\infty} \binom{n}{k} = \sum_{k=0}^{n} \binom{n}{k}$$

Now let look at definite series (1) where $F(n, k)$ is Gosper-summable. Since it's telescoping, we have

$$\sum_{k=-\infty}^{\infty} F(n, k) = \sum_{k=-\infty}^{\infty} [\, G(n, k+1) - G(n, k) \,] = 0$$

assuming that $G(n, k)$ has no singularities. Thus, we deduced that

$$F(n, k) \text{ is Gosper} - \text{summable} \Longrightarrow \sum_{k=-\infty}^{\infty} F(n, k) = 0$$

Conversely, if $F(n, k)$ has a finite support and it is a hypergeometric term, then if

$$\sum_{k=-\infty}^{\infty} F(n, k) \neq 0 \Longrightarrow F(n, k) \text{ is not Gosper} - \text{summable}$$

Wilf-Zeilberger's algorithm

Wilf-Zeilberger (or in short WZ) method is an application of Gosper's algorithm to definite summation, namely proving identities of the form

$$S_n \equiv \sum_{k=-\infty}^{\infty} F(n, k) = 1 \tag{2}$$

where F is a hypergeometric term with a finite support. As we saw in the section above, the summand F is not Gosper-summable. But what can we say about its difference wrt to n?

$$F(n+1, k) - F(n, k)$$

Suppose it's Gosper-summable, therefore there exists such $G(n, k)$ that

$$F(n+1, k) - F(n, k) = G(n, k+1) - G(n, k) \tag{3}$$

where G is a rational multiple of F. Summing (3) over all k, yields

$$\sum_{k=-\infty}^{\infty} [F(n+1,\, k) - F(n,\, k)] = \sum_{k=-\infty}^{\infty} [G(n,\, k+1) - G(n,\, k)]$$

The right-hand side is telescoping to zero, while the left hand side becomes $S_{n+1} - S_n$. This gives a linear recurrence equation of the first order

$$S_{n+1} - S_n = 0$$

Hence, S_n is a constant. Lastly, we need to make sure that $S_0 = 1$.

■

Don't ask:
what can the Computer do for me?

But rather:
what can I for the Computer?

The trend in mathematics these days is started to go from computer-assisted conjectures to computer-generated conjectures and then proofs.

computer-assisted conjectures. Pythagoras, Archimedes, Euler, Gauss, Riemann and all the other giants, who did extensive experimentation to find conjectures...

computer-generated conjectures. There exist powerful software packages that automatically finds conjectures, but without proving them. We will consider them later in the course.

computer-assisted proofs. Many proofs nowadays are computer-assisted, but in most of them computers are not mentioned. For example, *Coq* system.

computer-generated proofs. The first full-fledged computer-generated proofs started with WZ theory. In 1931, Kurt Godel proved that every consistent system of axioms is necessarily incomplete.

■ **Example 1** (Vandermonde's identity)

Prove

$$\sum_{k=0}^{n} \binom{a}{k}\binom{n}{k} = \binom{n+a}{a}$$

Proof.

We rewrite this identity in the form

$$\sum_{k=-\infty}^{\infty} F(n,\ k) = 1$$

where

$$F(n,\ k) = \frac{\binom{a}{k}\binom{n}{k}}{\binom{n+a}{a}}$$

Introduce

$$s_k = F(n+1,\ k) - F(n,\ k)$$

and apply Gosper's algorithm to s_k. If it's Gosper-summable, we find such $G(n,\ k)$ that

$$F(n+1,\ k) - F(n,\ k) = G(n,\ k+1) - G(n,\ k)$$

First we show that s_k is a hypergeometric term

$$\frac{s_{k+1}}{s_k} = ((k-a)(k-n-1)(1+k+a\,k+n-a\,n+k\,n)) \Big/ \left((k+1)^2(-a+k+a\,k-a\,n+k\,n)\right)$$

We choose

$$p_k = -a+k+a\,k-a\,n+k\,n$$

$$q_{k+1} = (k-a)(k-n-1)$$

$$r_{k+1} = (k+1)^2$$

A difference equation

$$q_{k+1}\ f_k - r_k\ f_{k-1} = p_k$$

$$(k-a)(k-n-1)\ f_k - k^2\ f_{k-1} = -a+k+a\,k-a\,n+k\,n$$

Its polynomial solution is a constant $f_k = -1$. Therefore, our function $G(n,\ k)$ is

$$G(n,\ k) = z_k = \frac{r_k}{p_k}\ f_{k-1}\, s_k = -\left(a!^2\, n!^2\right) \Big/ \left((k-1)!^2\,(a-k)!\ (a+n+1)!\,(n-k+1)!\right)$$

or

$$G(n,\ k) = \frac{(k-a-1)}{(a+n+1)}\ F(n,\ k-1)$$

Finally, we need to prove the initial case, namely that the sum

$$\sum_{k=0}^{n} \frac{\binom{a}{k}\binom{n}{k}}{\binom{n+a}{a}} = 1$$

is 1 for $n = 0$, which is indeed so:

$$\sum_{k=0}^{n} \frac{\binom{a}{k}\binom{n}{k}}{\binom{n+a}{a}} = \frac{\binom{a}{0}\binom{0}{0}}{\binom{0+a}{a}} = 1$$

Certificate

This pair of function $(G, \ F)$ is a WZ-pair

$$F(n+1, k) - F(n, k) = G(n, k+1) - G(n, k) \tag{4}$$

The rational function $R(n, k)$

$$R(n, k) = \frac{G(n, k)}{F(n, k)}$$

is called a certificate. Knowing $R(n, k)$ we can restore $G(n, k)$ and then verify identity (5). The latter is done by dividing (4) by $F(n, k)$

$$\frac{F(n+1, k)}{F(n, k)} - 1 = \frac{G(n, k+1)}{F(n, k)} - \frac{G(n, k)}{F(n, k)}$$

and making use

$$G(n, k) = R(n, k)\, F(n, k)$$

we obtain

$$\frac{F(n+1, k)}{F(n, k)} - 1 = \frac{F(n, k+1)}{F(n, k)}\, R(n, k+1) - R(n, k)$$

This defines a meaning of the certificate - we need to verify the above identity.

■ Example

Prove

$$\sum_{k=1}^{n} k \binom{n}{k} \binom{m}{k} = \frac{m\, n}{m+n} \binom{m+n}{m}$$

First we define the summand as a *Mathematica* function

```
F[n_, k_] := (k (m + n) Binomial[n, k] Binomial[m, k]) /
   (m n Binomial[n + m, m])
```

The problem is reduced to proving

$$\sum_{k=1}^{n} \frac{k\,(m+n) \binom{n}{k} \binom{m}{k}}{m\, n \binom{m+n}{m}} = 1$$

or

$$S(n) \equiv \sum_{k=1}^{n} F(n,\, k) = 1$$

This can be also written as

$$S(n) \equiv \sum_{k=-\infty}^{\infty} F(n,\, k) = 1$$

Check a single value $S(1)$

```
F[1, 1]
```

```
1
```

Therefore, if we can show that

$$S(n+1) - S(n) = 0 \tag{5}$$

then by induction we prove that $S(n) = 1$ for all n. Thus, we prove the original identity. By definition of $S(n)$, we have

$$S(n+1) - S(n) = \sum_{k=1}^{n+1} F(n+1,\, k) - \sum_{k=1}^{n} F(n,\, k) =$$

$$\sum_{k=-\infty}^{\infty} F(n+1,\ k) - \sum_{k=-\infty}^{\infty} F(n,\ k) = \sum_{k=-\infty}^{\infty} [F(n+1,\ k) - F(n,\ k)]$$

Note, $F(n+1,\ n+1)$ is not zero

```
F[n + 1, n + 1]
```

```
((1 + m + n) Binomial[m, 1 + n]) / (m Binomial[1 + m + n, m])
```

and therefore, we will have to make an additional step. We will take care of this at the end. Let us assume that $F(n+1,\ k) - F(n,\ k)$ is Gosper-summable, i. e. exist function G such that

$$F(n+1,\ k) - F(n,\ k) = G(n,\ k+1) - G(n,\ k)$$

and

$$\sum_{k=-\infty}^{\infty} [F(n+1,\ k) - F(n,\ k)] = \sum_{k=-\infty}^{\infty} [G(n,\ k+1) - G(n,\ k)] = 0 \qquad (6)$$

since the right hand side is telescoping under the additional assumptions

$$\lim_{x\to\infty} G(n,\ x) = 0$$

$$\lim_{x\to-\infty} G(n,\ x) = 0$$

We proceeed with Gosper's algorithm

```
s[k_] := F[n + 1, k] - F[n, k]
```

```
s[k + 1]
-------- // FunctionExpand // Factor
  s[k]
```

```
((k - m) (-1 + k - n) (k m + n + k n - m n)) / (k (1 + k) (-m + k m + k n - m n))
```

Since it's a hypergeometric term, we find a triple $(p_k,\ q_k,\ r_k)$ such that

$$\frac{s_{k+1}}{s_k} = \frac{p_{k+1}}{p_k} \frac{q_{k+1}}{r_{k+1}}$$

```
p[k_] := -m + k m + k n - m n
q[k_] := (k - m - 1) (k - n - 2)
r[k_] := k (k - 1)
```

Here is a difference equation for f_k

$$p_k = q_{k+1}\ f_k - r_k\ f_{k-1}$$

```
p[k] == q[k + 1] f[k] - r[k] f[k - 1]
```

```
-m - m n + k (m + n) == -(-1 + k) k f[-1 + k] + (k - m) (-1 + k - n) f[k]
```

The solution is obvious $f_k = -1$

```
Collect[p[k] - q[k + 1] f[k] + r[k] f[k - 1] /. f[k_] :> -1 , k, Factor]
```

```
0
```

Thus,

$$G(n,\ k) = \frac{r_k}{p_k}\ f_{k-1}\ s_k$$

```
G[n_, k_] := r[k]/p[k] (-1) s[k]
```

We need to show that $G(n,\ \pm\infty) = 0$.

```
G[n, x] // FunctionExpand // FullSimplify
```

```
-(m n (-1 + x) Gamma[m]^2 Gamma[n]^2) /
  (Gamma[1 + m + n] Gamma[1 + m - x] Gamma[2 + n - x] Gamma[x]^2)
```

```
Series[%22, {x, Infinity, 0}]
```

$$\left(\frac{1}{x}\right)^{m+n}\left(\frac{m\ n\ \mathrm{Gamma}[m]^2\ \mathrm{Gamma}[n]^2}{\pi^2\ \mathrm{Gamma}[1+m+n]} + O\left[\frac{1}{x}\right]^1\right)\ \mathrm{Sin}[m\,\pi - \pi\,x]\ \mathrm{Sin}[n\,\pi - \pi\,x]$$

To find a certificate we do the following

```
G[n, k]
------- // FunctionExpand // Simplify
F[n, k]
```

```
      (-1 + k) k
-----------------------
(-1 + k - n) (m + n)
```

```
                  (-1 + k) k
R[n_, k_] := --------------------
             (-1 + k - n) (m + n)
```

Verification of the proof by using the certificate

```
F[n + 1, k]         F[n, k + 1] R[n, k + 1]
----------- - 1 - ( ----------------------- - R[n, k] ) // FunctionExpand //
  F[n, k]                  F[n, k]
 FullSimplify
```

```
0
```

We noted at the beginning of this example that $F(n+1, n+1) \neq 0$, so formally

$$S(n+1) - S(n) = F(n+1, n+1) + \sum_{k=1}^{n} (F(n+1, k) - F(n, k))$$
$$= F(n+1, n+1) - G(n, 1) + G(n, n+1)$$

```
-G[n, 1] + F[n + 1, n + 1] + FunctionExpand[G[n, k]] /. k → n + 1
```

```
((1 + m + n) Binomial[m, 1 + n]) / (m Binomial[1 + m + n, m]) -
 (m n (1 + n) Gamma[m]^2) /
  ((-m - m n + m (1 + n) + n (1 + n)) Gamma[m - n] Gamma[1 + m + n])
```

```
FullSimplify[%]
```

```
0
```

■ Do-it-yourself

1) Prove

$$\sum_{k=0}^{n} \frac{n+2}{2k+1} \binom{n}{2k} \binom{2k+1}{k} 2^{n-2k-1} = \binom{2n+1}{n}$$

2) Show that the WZ algorithm fails on this identity:

$$\sum_{k=0}^{n} (-1)^k \binom{3k}{n} \binom{n}{k} = (-3)^n$$

References

[1] M. Petkovsek, H. Wilf, D. Zeilberger, *A = B*, Algorithms and Computations in Mathematics, AK Peters, 1996.

[2] R. L. Graham, D. E. Knuth, O. Patashnik, *Concrete Mathematics*, 2nd Ed., Addison-Wesley, 1994,

[3] W. Koepf, *Hypergeometric Summation. An Algorithmic Approach to Summation and Special Function Identities*, Vieweg, Braunschweig/Wiesbaden, 1998.